SYNTAX ZX80®

A PUBLICATION OF **THE HARVARD GROUP**

VOL.2 NO.1 JAN. 1981

16K RAM APPROVED; NEW INTERFACE

In mid-December, the Federal Communications Commission (FCC) finally approved Sinclair's 16K RAM for the U.S. market. Users with the North American model will soon be able to buy the expansion device and plug it into their machines. But Sinclair's Nigel Searle warns it could take several weeks before the part is available, for two reasons. First, the 16K RAM is produced in Britain, and demand in the U.K. already exceeds supply. But even if there were some waiting for shipment, a problem remains: the FCC wants some modifications to the British-model design.

There's other good news, too. C.A.I. Instruments, Midland, MI, has completed work on three interfaces. The basic interface hooks the ZX80 to either a serial or parallel printer. The other two are really options you can buy in addition to the first hookup. One will link up to a baud-rate generator; the other is an acoustic coupler, to interface with the phone.

Both products will soon be available on a mail-order basis through Sinclair. You can order the 16K RAM ($99.95) now. But Searle warns deliveries won't start for another few weeks. The interfaces, on the other hand, will be available in mid-January. Their prices range from $49.50 to $69.50 Mail orders to:

Sinclair Research, Ltd.
50 Staniford Street,
Boston, MA 02114
Or call: (203) 265-9171

NEW PRODUCT UPDATES

The ZX80 is having growing pains. Specifically, three features mentioned in the last issues are lagging behind schedule because the machine can't become powerful enough quickly enough.

The "Prestodigitizer", produced by Innovision, Los Altos, CA, allows you to write hand graphics on a tablet which appear on the screen as standard characters. Company President David Thornburg says he needs the new 8K ROM to finish adapting the machine to the ZX80. Sinclair had planned to have the new ROM available early in 1981, but that date has now been put back.

The development of another add-on, the "Cognivox", awaits the 16K RAM. Earlier, producer Voicetek of Goleta, CA, hoped to adapt the machine, which lets you talk with your computer, on the basis of the current 1K RAM. But Voicetek's Chris Georgiou now says he needs the 16K RAM to complete the work. Before FCC approval, however, it was illegal to import even a prototype. He says work can now proceed and the device should be ready by mid-January.

Last month we reported that MicroPeripheral Corp., of Mercer Island, WA, planned to market their ZX80 modem by January, 1981. Don Stoner of MicroPeripheral says the current target date is sometime in February or March because of software holdups. Most business for the modem will be mail order, but Stoner says there will be dealers in major American cities.

LIGHT CONTROL INTERFACE

This could be your last winter manually turning on lights, heaters, and appliances. Interface Technology, of Des Plaines, IL, already produces an interface connecting the BSR manual light control system to a TRS-80. Now, the company will adapt "The Microcommander" to the Sinclair ZX80. If all goes well, it should be ready within six months. The device interfaces directly to AC power lines. It sends digital codes synchronous with the power and gives the computer direct control of up to 256 lights and appliances.

Interface Technology's Ron Skinner says adapting the TRS-80 software to the Sinclair could be a bit tricky. Modifications will be needed to include a time-clock so "The Microcommander" can control on-off times for appliances like ovens and heaters. That's because the TRS-80 has more hardware, and derives some of its information from a source other than the Z80 chip. The Sinclair, on the other hand, runs solely off the Z80A chip.

Chips work serially, giving one message at a time. With an additional chip, the TRS-80 can both keep track of real time and switch on an appliance. The one-chip Sinclair and MicroAce machines require additional memory and specialized programming to perform both tasks.

Otis Imboden, author of last month's Compu-cal, wrote to tell of improvements in that program. We don't have space to print them here, but he will share them with other SYNTAX readers. The new Compu-cal works back to 4713 BC and now reports total days in positive numbers up to 999,999. Contact him at National Geographic Magazine, Washington, D.C., 20036.

BUBBLE SORT ROUTINE

"BUBBLE SORT" sorts numbers into ascending order. The sort routine starts on line 200; lines 1-190 provide random numbers to give the routine something to sort. The program generates 10 random numbers, prints them, puts them into ascending order and then prints them under the original numbers. You can use BUBBLE SORT with a string variable where the string variables are letters equal to 1 to 26 to put names or words in alphabetical order.

```
  1 REM * SET UP DIM ARRAY
 10 DIM A(10)
 30 FOR R=0 TO 10
 43 LET S=RND(100)
 45 LET A(R)=S
 50 NEXT R
```

OUR POLICY ON CONTRIBUTED MATERIAL

SYNTAX ZX80 invites you to express opinions related to the ZX80 and the newsletter. We will print, as space allows, letters discussing items of general interest. Of course, we reserve the right to edit letters to a suitable length and to refuse publication of any material.

We welcome program listings for all levels of expertise. Programs can be for any fun or useful purpose. We will test run each one before publishing it, but we will not debug programs; please send only workable listings.

In return for your listing, we will pay you a token fee of $2.00 per program we use. This payment gives us the nonexclusive right to use that program in any form, world-wide. This means you can still use it, sell it, or give it away, and so can we.

We will consider submissions of news and hardware or software reviews. Please keep articles short (50-100 words). Again, we reserve the right to edit accepted articles to a suitable length. We will pay 7 cents per 6 characters, including spaces and punctuation, for accepted articles.

If you want us to return your original program listing or article, please include a self-addressed, stamped envelope. Otherwise, we cannot return submitted material.

SYNTAX ZX80 is published monthly by Syntax, Inc., a wholly-owned subsidiary of The Harvard Group, RD 2, Box 457, Bolton Road, Harvard, MA 01451. Telephone 617 / 456-3661. Subscriptions: 12 issues, $25. Single issues, $4 each.

Publisher: Kirtland H. Olson
Editor: Ann L. Zevnik
Associate editor: Willa Marcus
Technical consultant: Eric K. Olson

Printed by Joseph E. Marhefka, Jr., Clinton Offset Printers, 472 High St., Clinton, MA, 01510.

BUBBLE SORT con't.

```
 100 REM * PRINT OUT ROUTINE
 105 LET P=0
 107 LET X=0
 110 PRINT A(X);" ";
 115 LET X=X+1
 120 IF X=10 THEN GO TO 150
 130 IF X<10 THEN GO TO 110
 150 LET P=P+1
 160 IF P=2 THEN GO TO 300
 190 REM **** BUBBLE SORT ROUTINE
 200 FOR I=0 TO 10
 210 LET J=9
 220 IF A(J-1)<A(J) THEN GO TO 2
60
 230 LET Y=A(J-1)
 240 LET A(J-1)=A(J)
 250 LET A(J)=Y
 260 LET J=J-1
 270 IF J>I THEN GO TO 220
 280 NEXT I
 290 GO TO 107
 300 STOP
```


BASIC SALARY SCHEDULING

Calculate salaries that increment on a straight-line basis using this scheduling program. This example uses teachers' salaries for a fictitious school based on education (bachelors, masters, and masters plus 30 hours credit) and years of experience. Lines 70, 80 and 85 specify the dollar increments.

After typing in the program, hit RUN (NL) to get the first screenful of data. Then press CONT twice (you will get the program listing after pressing CONT once, but just press it again) then (NL) to see the rest of the data. (Line 5—SCHOOL is 1 word. Line 7—put a space between SALARY and SCHEDULE.)

```
   5 PRINT "       CIRCLEVILLE S
CHOOLS"
   7 PRINT "     1980-81 SALARY
SCHEDULE"
  10 PRINT "YRS","BACH","MAST","
M+30"
  20 LET N=0
  30 LET B=10000
  40 LET M=11100
  45 LET M1=12000
  50 PRINT N,B,M,M1
  60 LET N=N+1
  70 LET B=B+227
  80 LET M=M+281
  85 LET M1=M1+311
  90 IF N=12 THEN GO TO 500
 100 PRINT N,B,M,M1
 110 GO TO 60
 500 PRINT N,B,M,M1
 510 LET N=N+1
 520 IF N=19 THEN GO TO 700
 530 LET M=M+281
 540 PRINT N,,M,M1
 550 GO TO 510
 700 LET M1=M1+311
 710 LET N=N+1
 720 IF N=26 THEN GO TO 900
 730 PRINT N,,,M1
 740 GO TO 710
 900 STOP
```

Robert DeMunbrun, Rushville, IN

CHART YOUR BIORHYTHMS

This biorhythm program for 8K ROM machines graphically displays your emotional, intellectual and physical cycles (represented by E, I, and P) based on your birthday. The cycles move in regular sine waves with differing periods, starting at 0 on your birthday. When they cross 0, prepare yourself for critical days.

To use, enter RUN (NL). The program is too large for prompts, so enter this information: month (NL), day (NL) and year (NL) of birth, then month (NL), day (NL), and year (NL) of plot start.

The machine will display a continuously scrolling graph. The center line represents critical times. The left side shows recuperative times; the right side extra energy times. Numbers to the left of display indicate days elapsed from start date. To slow the display, hold the P (for pause) key. Plot dates can range from 1901 to 2099 AD.

```
 103 DIM N(3)
 106 LET P2=2*PI
 107 LET W=10
 108 LET K=5
 109 LET L=7
 140 FOR T=1 TO 2
 145 INPUT M
 160 INPUT D
 165 INPUT Y
 170 LET N(T)=INT (365.25*(Y-(M<
3)))+INT (30.6001*(M+1+12*(M<3))
)+D
 192 NEXT T
 230 LET N =N(2)-N(1)
 290 PRINT AT L,0;B=N(1)-N(2)+1;
AT 1,(SIN (P2*B/23)*W+W+K);"P";A
T L,(SIN (P2*B/28)*W+W+K);"E";AT
 L,(SIN (P2*B/33)*W+W+K);"I";AT
L,(W+K);":"
 295 SCROLL
 297 PAUSE 20
 298 IF INKEY$="P" THEN PAUSE 1
00
 300 LET B=B+1
 310 GO TO 290
```

COMPUTER MODELS

One of the most useful capabilities of a computer is its ability to calculate many cases for a business or technical problem and display the results for you to analyze. This is one kind of computer model.

Let's take a simple case and work it through to see how you construct a model of your own. For our example, we will calculate the profit for various quantities of a product.

Overall, we need to calculate the revenue from sales and the cost of goods sold. Total revenue (TREV) equals sales price for each unit (USP) times number of units sold (USLS). Total cost is the sum of initial fixed costs (IFXCOST) and variable cost per unit (UVCOST) times number of units sold. Thus we can write:

```
1000 LET TREV=USP*USLS
2000 LET VARCOST=UVCOST*USLS
2500 LET TCOST=VARCOST+IFXCOST
3000 LET PROFIT=TREV-TCOST
5000 PRINT PROFIT
```

Equations define the relationships among the variables, but the model program cannot run because costs and prices are not yet assigned. Now we must specify these values:

```
  50 LET USLS=1000
 100 LET USP=12
 200 LET UVCOST=2
 210 LET IFXCOST=4000
```

Now you can run the program and see the answer. Big deal, this you could do with any cheap four-banger. But your ZX80 can automate the process for you. Here's how.

Delete line 50 and add a FOR-NEXT loop at line 900. We will find the profit for unit sales from 200 to 1000 pieces in 100-unit increments. Insert the NEXT statement after the PRINT

statement to see all the values:

```
 900 FOR I=2 TO 10
 920 LET USLS=100*I
5900 NEXT I
```

From here forward, you can run the program and see the developing result after each block of instructions. Try it.

Change the PRINT statement so you can see the unit sales and the profit for each:

```
5000 PRINT USLS,PROFIT
```

For convenience, display the values of prices and costs. Put the statement before the loop so it prints only once. For better appearance, use an empty PRINT line to get a space:

```
 790 PRINT
 800 PRINT "UNIT SALES PRICE ";U
SP
 810 PRINT "UNIT VARIABLE COST "
;UVCOST
 820 PRINT "FIXED COSTS ";IFXCOS
T
```

Now the screen displays all the data we put in as well as the results as sales vary. We are using computer capability.

Let's further suppose that producing more than 800 units needs more space, so we suffer an increase of fixed expense of UPFX. Implement this by adding:

```
 220 LET UPFX=1000
2100 LET FXCOST=IFXCOST
2400 IF USLS>800 THEN LET FXCOST
=FXCOST+UPFX
```

and modify line 820 to:

```
 820 PRINT "FIXED COSTS ";IFXCOS
T;",";IFXCOST+UPFX
```

So far, we've just varied the number of units sold, so let's now change the sales price at the same time. Since sales price will now vary, we need to change line 800. We want to try three sales prices, say 10, 12 and 14; we'll need to add another loop. Delete lines 100 and 800:

```
  75 DIM A(3)
  85 LET A(1)=10
  86 LET A(2)=12
  87 LET A(3)=14
 940 PRINT USLS,
 950 FOR J=1 TO 3
 960 LET USP=A(J)
5000 PRINT PROFIT,
5100 NEXT J
```

Now we need labels atop the table:

```
 840 PRINT
 850 PRINT "UNIT",,"PROFIT"
 860 PRINT "SALES","USP=";A(1),A
(2),A(3)
 870 PRINT
```

Lines 1000 to 3000 express standard relations for break-even analysis of any business. All the remaining model parts fix values, vary conditions, or tabulate and display results. You can alter this program by changing values, by changing relationships, by making costs or revenue depend on sales in more complex ways, or by changing the output to provide another data summary.

Apply these principles to any model situation that helps you make personal or business choices.

RAM addresses start at 16384 and end at 17408 for the 1K machine.

ATTENTION M.L. PROGRAMMERS:

In conjunction with our upcoming series on machine language programming, we need machine language program listings. Please subnit original programs or translations of our proviously published listings. We will pay our usual rates for listings and accompanying explanatory text. Become famous!

MEMORY WINDOW

Here's a simple utility program that displays 16 consecutive bytes of memory in hexidecimal, decimal, and character formats. It requires a 4 hex digit starting address (upper limit 7FEF or 32751). Addresses less than 4 digits or larger than 7FEF can result in an error. Non-hex input is ignored. For example, try 40AA to display lines 120 and 130.

```
  10 CLEAR
  20 DIM A(3)
  30 DIM B(3)
  40 LET C=0
  50 INPUT H$
  60 CLS
  70 FOR J=0 TO 3
  80 IF CODE(H$)<28 THEN GO TO 1
0
  90 IF CODE(H$)>43 THEN GO TO 1
0
 100 LET C=C+(CODE(H$)-28)*(16**
(3-J))
 110 LET H$=TL$(H$)
 120 NEXT J
 130 FOR I=0 TO 15
 140 LET D=C
 150 LET E=PEEK(D)
 160 LET F=E
 170 FOR J=0 TO 3
 180 LET A(J)=D/(16**(3-J))
 190 LET D=D-(16**(3-J)*A(J))
 200 LET B(J)=F/(16**(3-J))
 210 LET F=F-(16**(3-J)*B(J))
 220 NEXT J
 230 LET C=C+1
 240 FOR J=0 TO 3
 250 PRINT CHR$(A(J)+28);
 260 NEXT J
 270 PRINT,
 280 FOR J=2 TO 3
 290 PRINT CHR$(B(J)+28);
 300 NEXT J
 310 PRINT ,E,
 320 PRINT CHR$(E)
 330 NEXT I
 340 GO TO 10
```

Neil Marshall,
Gaithersburg, MD

HOW MANY BYTES TO THE LINE?

This subroutine determines the exact number of bytes used for each line of program instruction as well as an entire program.

Enter a decimal starting address after the first prompt (the lowest line number always starts at 16426). The program shows the starting address and contents of each line. Subtract one starting address from the next to see how many bytes that line uses. Press (NL) to see each screenful of information. At the end, it will display the number of instructions, total bytes, average bytes and the starting address available for the next program. This routine assumes it will find a code 248 (STOP), so make sure each program ends with a STOP.

To exit, enter E (NL) when it asks for next program. To analyze another program, type it in ahead of this one with smaller line numbers. Type RUN 9200 (NL). To go on to another program, hit (NL) after NEXT PROGRAM, then enter the starting address given for the next program.

```
9200 REM DUMP INSTRUCTIONS
9205 PRINT "START AT?"
9210 INPUT S
9215 LET B=S
9220 LET C=0
9225 CLS
9230 FOR I=1 TO 15
9235 PRINT S,
9240 LET C=C+1
9245 IF PEEK(S)=118 THEN GO TO
9270
9250 PRINT CHR$(PEEK(S));
9255 IF PEEK(S)=248 THEN GO TO
9310
9260 LET S=S+1
9265 GO TO 9245
9270 PRINT
9275 LET S=S+3
9280 NEXT I
9290 INPUT Z$
9295 IF Z$="E" THEN GO TO 9390
9300 GO TO 9225
```

```
9310 PRINT
9315 PRINT
9320 PRINT C,"INSTRUCTIONS"
9325 LET A=S-B+1
9330 PRINT A,"BYTES"
9335 LET D=A/C
9340 PRINT D,"AVERAGE BYTES"
9345 LET S=S+4
9350 PRINT S,"NEXT PROGRAM"
9355 INPUT Z$
9360 IF Z$="E" THEN GO TO 9390
9370 IF Z$="" THEN GO TO 9215
9375 GO TO 9205
9390 PRINT "END DUMP"
9392 GO TO 9399
9395 RETURN
9399 STOP
```

Martin H. Irons, Goshen, NY

INTERFACE TO THE REAL WORLD

Your ZX80 or MicroAce can talk to other machines if you wire and program a simple output port. The port latches any 8-bit data byte output from the data bus and is internally buffered. You can sink up to 15 mA per line to drive LED's or relay coils (to access appliances). You can also (as this project does) source up to 15 mA per line. This device typifies memory-mapped I/O, the building block to connect other add-ons.

You talk to your interface via software using the BASIC POKE function. It is completely reset on system power-up. The location 12288(decimal)=3000H addresses the port device. POKE the data to location 12288 each time you want to change the output. Then your program can perform additional, unrelated tasks while the latch remembers your previous order.

The program listed sequences through 8 LED's at a rate you select. Repeating displays give a moving effect like a theatre sign. With minor software changes, you may sweep it in either direction and any format. Modify the delay between each LED being gated-on (and thus the rate of motion) by altering line 120. For example, decrease the FOR-TO number to 25 for a faster display.

This scheme works by making the I/O chip look like a memory location. For this output-only device, you can use a ROM address because the ZX80 never writes to a ROM address for any other purpose. I/O ports need empty addresses.

Decoder logic generates the address from A13 "ANDed" with A12 and $\overline{\text{MREQ}}$ to select the 8212 I/O chip. Since the address is not fully decoded, the port responds to every address between 3000H and 3FFFH. Data clocked-in via the $\overline{\text{WR}}$ pulse is latched to the outports on the rising edge of $\overline{\text{WR}}$.

The connector on the rear of the ZX80 chassis provides access. The +5 Volt, on-board, regulated supply operates your port device. A 7400 quad 2-input NAND gate and the 74LS02 quad 2-input NOR gate decode the address 12288.

Two operating modes are possible with the 8212. As shown, you have an output port to take signals from the ZX80 to the world outside. But, wired and addresses differently, the 8212 could take data from outside your ZX80 and make it available to the computer for processing or display. For more info on other connections see Adam Osborne's Introduction to Microcomputers, Volume II, June 1977 revision, pp 4-122 to 4-131.

Project parts are readily available from electronics suppliers listed in the back of electronics hobby magazines. One supplier is Jameco Electronics, 1021 Howard Ave., San Carlos, CA, 94070, (415)592-8097. Prices: $3.25 for the 8212 I/O IC, 29¢ for a TTL IC. Any type or color LED (about 20¢ each) is OK.

In the diagram (p.8) 14B means pin 14, bottom side, of the ZX80 connector; 14T means pin 14, top side. Count pins from left to right with keyboard toward you and readable. On the board, pin 23T is labeled 23A.

```
  10 REM * EXAMPLE OF MEMORY-MAP
PED OUTPUT *
  20 REM * USING AN INTEL 8212 P
ARALLEL INPUT/OUTPUT DEVICE. *
  40 REM * DELAY LOOP AT STATEME
NTS 120 AND 130. *
  50 REM * CONTROLS THE STROBE/M
ULTIPLEXER RATE. *
  60 LET N=0
  70 LET D=2**N
  80 LET N=N+1
  90 POKE 12288,D
 100 LET A=128-D
 120 FOR I=1 TO 100
 130 NEXT I
 140 IF A=0 THEN GO TO 60
 150 GO TO 70
 160 REM * 12288=ADDRESS (3000H)
 170 REM OF I/O PORT (8212)
 180 REM * PROGRAM CONTINUES*
 190 REM * UNTIL "BREAK" IS *
 200 REM * DEPRESSED. *
 210 STOP
```



DEAR EDITOR:

I am happy to see a publication for ZX80 users. I hope you won't forget about us computer enthusiasts north of the border! I would appreciate the inclusion of Canadian hardware and software suppliers. Also, I would like to hear from anyone in this area with a ZX80 to discuss applications and problem-handling.

Larry Boyd, Guelph, Ontario
519/836-8136

Gladstone Electronics in Toronto and Scarborough is the Canadian agent for Sinclair and sells ZX80s both in-store and by mail for $329.95 Canadian. Gladstone also plans to offer Sinclair products through selected Canadian computer stores. They offer a list of ZX80 software, but I don't know if it contains Canadian suppliers. They will publish a free newsletter irregularly for their customers, and want to start a users group. For information, contact Howard Gladstone, 1736 Avenue Rd., Toronto, Ontario, Canada, M5M 3Y7, 416/787-1448.—AZ

I recently tried to use the Cryptoquote program from the first issue but cannot get it to run. Every time I enter F after the command prompt, the machine gives a 2/480 error code. N=20 at that point, so I don't understand why the error indicates a variable name not found at that line.

In a program I wrote, I found the machine giving a 2/X error at a line number that did not exist. Is the machine malfunctioning?

John La Pin, Germantown, WI

Cryptoquote is error-free. Check that you didn't mistype any variable names. Or, you may have typed O instead of 0 (zero) in line 480. If your own program is too large for memory, the ZX80 will truncate the line number in the error code, using only as much space as it has. This may appear to be a different line number.—AZ

When I plug my ZX80 into a TV (I've tried 4 different models), I get a distracting interference pattern, making the display hard to read. I tried to filter it out with a capacitor with almost no effect. Is there something wrong with my machine?

Paul Byrne, Wichita, KS

Rod Rakes wrote to tell us that squeezing the case removes most hash lines. This shows that the silver fingers under the PC board aren't contacting the metallized case. Apply pressure to the plastic case near the rear. Rod also found that bending the video cable helps. According to David Ornstein of Sinclair, if TV interference continues to plague you, you can return your ZX80 for replacement.—AZ

I have been playing with my ZX80 for nearly a month and I'd like suggestions of tape recorders for saving programs. We have two and neither will work. I now find that the best puts out barely >3V at its highest level. My only solution right now (besides blowing $50 on a new recorder which may or may not work) is to build a very small amplifier.

Steve Crandall, Stony Brook, NY

We use a Radio Shack CTR-80 recorder with few problems. If you don't want to buy a new recorder, write to Sinclair in Boston and ask for their improved loading instructions. Better still, call their technical information number, 617/367-2555, for a faster reply.—AZ

MACHINE COMPATIBILITY

A beginner wrote to ask: All of the following computers have the designation 80—ZX80, Z80, Z80A, TRS-80. Are programs or software usable between each of these computers? Also there is a Z80 Softcard to apply Z80 programs to the Apple II. Is it reversible?

First of all, please note the distinction between processors and computers. Z80 and Z80A are the central processors providing the brains to the TRS-80 and ZX80 computer, respectively. The 2 processors have no high level language (such as BASIC) compatibility, but do understand each other in machine language because their instruction sets are the same.

So machine language routines, or programs, written without using the operating system are usable on both machines. (The operating system controls where things are stored in memory and handles input from the keyboard and storage and output to viedo and storage.) Unfortunately, such programs are limited to computations or processing (manipulating bits), which doesn't help beginners.

The TRS-80 and ZX80 speak similar but not identical forms of BASIC and use different character sets. Think of it as different dialects of the same language, using different alphabets. They cannot execute identical BASIC programs. To use the same programs, you must either put an interface between them that can translate or program 1 machine to translate the other's BASIC. Alternatively, you can translate programs yourself.

If you write your own translations, watch out for instructions beyond the ability of the machine. For instance, TRS-80 can handle multiple statements per line, which the ZX80 cannot. Or one machine may not use certain commands, like IF...THEN ELSE. We'll have an article on program translation in a later issue.

The Apple Microsoft card contains a Z80 processor which makes the Apple emulate a Z80 machine. Some but not all Z80 software will run on an Apple. In any event, the card is only usable on Apple computers.

REM STATEMENTS

REM statements can be useful tools for beginning programmers. REM stands for REMark. You can also think of it as REMinder or REMember.

The computer does not execute REM statements. It will ignore anything you write after REM in any line. Use REM to write notes to yourself about the functions of various parts of your program. For example, if you wrote a program that asked for data, performed a computation, then displayed the results, you could insert REM statements delineating these parts:

```
 10 REM THIS PROGRAM CALCULATES
AVERAGE INCOME FOR 10 YEARS.
 20 REM NEED 10 YEARLY INCOME F
IGURES.
```

The next lines would print requests for yearly incomes and input that information. Then you get to the calculation:

```
100 REM NOW CALCULATE AVERAGE
```

The next lines tell the computer to add the incomes and divide by the number of years.

Finally, you want to see the results:

```
150 REM PRINT ANSWER
```

The next line is a print statement to tabulate the average yearly income. >I2

Learn how to use a computer for $249. And keep the computer.

Introducing the Computer Learning Lab, with the Sinclair ZX80 personal computer.

The Sinclair ZX80 is a powerful, personal computer already in use in tens of thousands of homes and businesses.

Now, for just $249*, the Computer Learning Lab will teach you and your children how to use this complete, expandable computer right in your own home. And when you're done, you get to keep the computer!

The Computer Learning Lab is a self-paced course that teaches the ins and outs of computers. It cuts away jargon and mystique, taking you straight into BASIC—the most common, easy-to-use computer language. And the computer itself does the teaching.

100 LESSONS MAKE LEARNING EASY.

You just take the ZX80 out of the box, connect it to your TV and an ordinary cassette recorder (connectors are provided), and slip in the pre-programmed Learning Lab cassettes. There's nothing extra to purchase.

You'll be working with the computer your very first day!

The cassettes take you through 100 experiments that teach you how to solve problems with the ZX80.

You learn by doing. By actually working with the computer.

And the lessons are designed to be fun and involving. You create your own programs for games, code breaking, interest calculation, and other topics. Then you can apply the principles you've learned to more complex problems.

In fact, you'll be a master of the ZX80. To use in your business, for home budgeting, or just for fun.

We'll also send you a catalog full of ready-to-use programs—from Loan Amortization to Lunar Lander—available for as little as $6.95.

THE ADVANCED DESIGN OF THE ZX80.

The ZX80 is the world's first truly portable computer. It features a touch-sensitive keyboard and a 32-character by 24-line display.

And it performs like a much larger and more expensive computer. Single keystroke entries make typing programs fast and easy. An automatic error detection feature tells you if you make mistakes. And program editing helps you correct them. Yet the complete Computer Learning Lab, including computer, is still several hundred dollars less expensive than any comparable computer alone.



The Computer Learning Lab is a family learning aid. Children 10 and above will quickly understand the principles of computing—and have fun learning.

The ZX80 is backed by a 30-day money-back guarantee and a 90-day limited warranty with a national service-by-mail facility. Extended service contracts for the ZX80 are available for a minimal charge.

The complete package includes the ZX80, a 128-page guide to computing, a workbook, six program cassettes, and two blank cassettes for storing your own programs.

sinclair

MASTERCHARGE OR VISA ORDERS CALL: (203) 265-9171. We'll refund the cost of your call.
INFORMATION: General and technical (617) 367-1898 or 367-1988. Phones open Mon.—Fri. 8 AM to 8 PM EST.

To: SINCLAIR RESEARCH LTD., 475 MAIN ST., P.O. BOX 3027, WALLINGFORD, CT 06492.

Please send me _____ Computer Learning Lab(s) with the ZX80 personal computer at $249 each (U.S. dollars).

Please send me _____ ZX80 computer(s) only, without the Learning Lab, at $199.95* each.

I enclose check/money order payable to Sinclair Research Ltd. for $_____ (Add $5.00 for shipping.)

*For Conn. deliveries, add sales tax.

Name _____

Address _____

City _____ State _____ Zip _____

Occupation _____ Age _____

Intended use of ZX80: _____ SA-1-1

Have you ever used a computer? ☐ Yes ☐ No Do you own another personal computer? ☐ Yes ☐ No

> REM STATEMENTS con't.

This trick is especially useful to remind yourself what you wanted to do at each step of your program. It also helps others who might try to debug a program that won't run or adapt one you wrote.

You can also use REM statements to include copyright notices in programs you write, or to insert disclaimers of responsibility, as for games like rolling dice that may be used for gambling. I don't know if this constitutes legal protection, but it does make your intent clear.

LIGHTS OF THE CITY

This program uses the 8K ROM's graphics and animation capabilities to generate an ever-changing video pattern like a growing city. To use, type RUN (NL) and then enter a pause time (number of frames to display between movements). Good times are 7 and 10. The image may jump around a bit, but changing the pause time will help. To "fast forward," hold down the P key. The computer will build the city but not display it until you release P.

The program defines two variables, X & Y, to indicate line & column of the current position. It selects a random direction (up, down, right, left) and changes the current position accordingly, displaying a character there. The display continues for Z frames then changes. The program runs until it fills memory or you hit BREAK.

```
  2 RANDOMIZE
  5 INPUT Z
 10 LET X=12
 15 LET Y=16
 20 GO TO 31+40*RND
 40 LET X=X+(X<>21)
 41 LET C$="▄"
 42 GO TO 100
 50 LET Y=Y+(Y<>31)
 51 LET C$="▌"
 52 GO TO 100
 60 LET X=X-(X<>0)
 61 LET C$="▀"
 62 GO TO 100
 70 LET Y=Y-(Y<>0)
 71 LET C$="▐"
100 PRINT AT X,Y;C$
103 IF INKEY$="P" THEN GO TO 20
105 PAUSE Z
110 GO TO 20
```

PRINT 17408-PEEK(16400)-PEEK(16401)*256 as the last line of your program will print the number of bytes still available.

SYNTAX
THE HARVARD GROUP
Bolton Road, Harvard, Mass. 01451